

Geodesic IVP

Our game here is to find the solution of

$$0 = \frac{d^2 x^a}{ds^2} + \Gamma_{bc}^a(x) \frac{dx^b}{ds} \frac{dx^c}{ds}$$

subject to the initial conditions $x^a(s) = x^a$ and $dx^a(s)/ds = \dot{x}^a$ at $s = 0$

Algorithm

By successive differentiation of the above equation we can compute

$$\frac{d^n x^a}{ds^n} = -\Gamma_{bcd_n}^a \frac{dx^b}{ds} \frac{dx^c}{ds} \frac{dx^{d_n}}{ds}$$

at $s = 0$. The $\Gamma_{bcd_n}^a$ are the *generalised connections*.

We can then construct the Taylor series solution for $x^a(s)$

$$x^a(s) = x^a - \sum_{k=2}^{\infty} \frac{1}{k!} \Gamma_{bcd_k}^a \dot{x}^b \dot{x}^c \dot{x}^{d_k}$$

```
# =====
# Compute the generalised connection, for later use in series expansion of x^a(s)
# =====

::KeepHistory(false).
::PostDefaultRules( @@collect_terms!(%), @@sumflatten!(%) ).

{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#,v#}::Indices.

\nabla_{#}::PartialDerivative.
\partial_{#}::PartialDerivative.

g_{a b}::Metric.
g^{a b}::Metric.
\delta^{a}_{b}::KroneckerDelta.
```

```

R_{a b c d}::RiemannTensor.
R^{a}_{b c d}::RiemannTensor.

# --- the Q are shorthand for the genGamma, saves typing -----
Q^{a}_{b c},Q^{a}_{b c d},Q^{a}_{b c d e},Q^{a}_{b c d e f},Dx^{a}::SortOrder.

Q^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).
Q^{a}_{b c d}::TableauSymmetry(shape={3}, indices={1,2,3}).
Q^{a}_{b c d e}::TableauSymmetry(shape={4}, indices={1,2,3,4}).
Q^{a}_{b c d e f}::TableauSymmetry(shape={5}, indices={1,2,3,4,5}).
Q^{a}_{b c d e f g}::TableauSymmetry(shape={6}, indices={1,2,3,4,5,6}).

\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

# =====
# recursively compute  $d^n x^a/ds^n$  at  $s = 0$ 
# =====

# ---  $A^a = x^a$ ,  $B^a = dx^a/ds$  at  $s = 0$ 

deriv00:=A^{a};
deriv01:=B^{a};

deriv02:=-\Gamma^{a}_{b c} B^{b} B^{c};

deriv03:=\nabla{@(deriv02)}:
@prodrule! (%):
@substitute! (%)(\nabla{B^{a}}->@(deriv02)):
@substitute! (%)(\nabla{\Gamma^{m}_{s t}}->B^{d}\partial_{d}\{\Gamma^{m}_{s t}\}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%);

deriv04:=\nabla{@(deriv03)}:
@distribute! (%):
@prodrule! (%):
@substitute! (%)(\nabla{B^{a}}->@(deriv02)):
@substitute! (%)(\nabla{\Gamma^{m}_{s t}}->B^{d}\partial_{d}\{\Gamma^{m}_{s t}\}):
@substitute! (%)(\nabla{\partial_{e}\{\Gamma^{m}_{s t}\}}->B^{d}\partial_{d e}\{\Gamma^{m}_{s t}\}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%);

```

```

deriv05:=\nabla{@(deriv04)}:
@distribute! (%):
@prodrule! (%):
@substitute! (%)(\nabla{B^{a}}->@(deriv02)):
@substitute! (%)(\nabla{\Gamma^m_{s t}}->B^d\partial_{d}\{\Gamma^m_{s t}\}):
@substitute! (%)(\nabla{\partial_e\{\Gamma^m_{s t}\}}->B^d\partial_{d e}\{\Gamma^m_{s t}\}):
@substitute! (%)(\nabla{\partial_{ef}\{\Gamma^m_{s t}\}}->B^d\partial_{d ef}\{\Gamma^m_{s t}\}):
@prodsort! (%): @rename_dummies! (%): @canonicalise! (%);

# --- deriv06 to leading order in s equals  $-\Gamma^a_{bcdefg}$  which vanishes
deriv06:=0;

```

$$deriv00 := A^a$$

$$deriv01 := B^a$$

$$deriv02 := (-1) \Gamma^a_{bc} B^b B^c$$

$$deriv03 := -B^b B^c B^d \partial_b \Gamma^a_{cd} + 2 B^b B^c B^d \Gamma^a_{be} \Gamma^e_{cd}$$

$$\begin{aligned}
deriv04 := & B^b B^c B^d B^e \Gamma^f_{bc} \partial_f \Gamma^a_{de} + 4 B^b B^c B^d B^e \Gamma^f_{bc} \partial_d \Gamma^a_{ef} - B^b B^c B^d B^e \partial_{bc} \Gamma^a_{de} \\
& - 2 B^b B^c B^d B^e \Gamma^a_{fg} \Gamma^f_{bc} \Gamma^g_{de} - 4 B^b B^c B^d B^e \Gamma^a_{bf} \Gamma^f_{cg} \Gamma^g_{de} + 2 B^b B^c B^d B^e \Gamma^a_{bf} \partial_c \Gamma^f_{de}
\end{aligned}$$

$$\begin{aligned}
deriv05 := & -2 B^b B^c B^d B^e B^f \Gamma^g_{bc} \Gamma^h_{dg} \partial_h \Gamma^a_{ef} - 6 B^b B^c B^d B^e B^f \Gamma^g_{bc} \Gamma^h_{de} \partial_g \Gamma^a_{fh} + B^b B^c B^d B^e B^f \partial_g \Gamma^a_{bc} \partial_d \Gamma^g_{ef} + 3 B^b B^c B^d B^e B^f \Gamma^g_{bc} \partial_{dg} \Gamma^a_{ef} \\
& - 12 B^b B^c B^d B^e B^f \Gamma^g_{bc} \Gamma^h_{dg} \partial_e \Gamma^a_{fh} - 6 B^b B^c B^d B^e B^f \Gamma^g_{bc} \Gamma^h_{de} \partial_f \Gamma^a_{gh} + 6 B^b B^c B^d B^e B^f \partial_b \Gamma^a_{cg} \partial_d \Gamma^g_{ef} \\
& + 6 B^b B^c B^d B^e B^f \Gamma^g_{bc} \partial_{de} \Gamma^a_{fg} - B^b B^c B^d B^e B^f \partial_{bcd} \Gamma^a_{ef} + 12 B^b B^c B^d B^e B^f \Gamma^a_{gh} \Gamma^g_{bc} \Gamma^h_{di} \Gamma^i_{ef} - 6 B^b B^c B^d B^e B^f \Gamma^a_{gh} \Gamma^g_{bc} \partial_d \Gamma^h_{ef} \\
& + 4 B^b B^c B^d B^e B^f \Gamma^a_{bg} \Gamma^g_{hi} \Gamma^h_{cd} \Gamma^i_{ef} + 8 B^b B^c B^d B^e B^f \Gamma^a_{bg} \Gamma^g_{ch} \Gamma^h_{di} \Gamma^i_{ef} - 8 B^b B^c B^d B^e B^f \Gamma^a_{bg} \Gamma^h_{cd} \partial_e \Gamma^g_{fh} \\
& - 4 B^b B^c B^d B^e B^f \Gamma^a_{bg} \Gamma^g_{ch} \partial_d \Gamma^h_{ef} - 2 B^b B^c B^d B^e B^f \Gamma^a_{bg} \Gamma^h_{cd} \partial_h \Gamma^g_{ef} + 2 B^b B^c B^d B^e B^f \Gamma^a_{bg} \partial_{cd} \Gamma^g_{ef}
\end{aligned}$$

$$deriv06 := 0$$

```

# =====
#
# Now we need to expand deriv0* as a Taylor series around x=0
# These will be used in constructing the general solution of the geodesic equation near x=0, i.e. x^a(s)
# To obtain a series accurate to terms including O(L^6) we will need the following derivatives at x=0
#
#   deriv02,a   deriv02,ab   deriv02,abc   deriv02,abcd
#   deriv03,a   deriv03,ab   deriv03,abc
#   deriv04,a   deriv04,ab
#   deriv05,a
#
# =====

# --- Taylor series for deriv02 -----

@distributed!(deriv02):
term00:=@deriv02:

tmp:=A^a \partial_a{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_a{B^b} -> 0):
term01:=@tmp:

tmp:=A^a \partial_a{@(term01)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_a{B^b} -> 0, \partial_a{A^b} -> 0):
term02:=@tmp:

tmp:=A^a \partial_a{@(term02)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_a{B^b} -> 0, \partial_a{A^b} -> 0):
term03:=@tmp:

tmp:=A^a \partial_a{@(term03)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_a{B^b} -> 0, \partial_a{A^b} -> 0):
term04:=@tmp:

@substitute!(term00)(\partial_{a b c d}{\Gamma^p}_{m n} -> Q^p_{m n a b c d}):

```

```

@substitute!(term00)(\partial_{a b c}\Gamma^p_{m n} -> Q^p_{m n a b c}):
@substitute!(term00)(\partial_{a b}\Gamma^p_{m n} -> Q^p_{m n a b}):
@substitute!(term00)(\partial_a\Gamma^p_{m n} -> Q^p_{m n a}):
@substitute!(term00)(\Gamma^p_{m n} -> Q^p_{m n}):

@substitute!(term01)(\partial_{a b c d}\Gamma^p_{m n} -> Q^p_{m n a b c d}):
@substitute!(term01)(\partial_{a b c}\Gamma^p_{m n} -> Q^p_{m n a b c}):
@substitute!(term01)(\partial_{a b}\Gamma^p_{m n} -> Q^p_{m n a b}):
@substitute!(term01)(\partial_a\Gamma^p_{m n} -> Q^p_{m n a}):
@substitute!(term01)(\Gamma^p_{m n} -> Q^p_{m n}):

@substitute!(term02)(\partial_{a b c d}\Gamma^p_{m n} -> Q^p_{m n a b c d}):
@substitute!(term02)(\partial_{a b c}\Gamma^p_{m n} -> Q^p_{m n a b c}):
@substitute!(term02)(\partial_{a b}\Gamma^p_{m n} -> Q^p_{m n a b}):
@substitute!(term02)(\partial_a\Gamma^p_{m n} -> Q^p_{m n a}):
@substitute!(term02)(\Gamma^p_{m n} -> Q^p_{m n}):

@substitute!(term03)(\partial_{a b c d}\Gamma^p_{m n} -> Q^p_{m n a b c d}):
@substitute!(term03)(\partial_{a b c}\Gamma^p_{m n} -> Q^p_{m n a b c}):
@substitute!(term03)(\partial_{a b}\Gamma^p_{m n} -> Q^p_{m n a b}):
@substitute!(term03)(\partial_a\Gamma^p_{m n} -> Q^p_{m n a}):
@substitute!(term03)(\Gamma^p_{m n} -> Q^p_{m n}):

@substitute!(term04)(\partial_{a b c d}\Gamma^p_{m n} -> Q^p_{m n a b c d}):
@substitute!(term04)(\partial_{a b c}\Gamma^p_{m n} -> Q^p_{m n a b c}):
@substitute!(term04)(\partial_{a b}\Gamma^p_{m n} -> Q^p_{m n a b}):
@substitute!(term04)(\partial_a\Gamma^p_{m n} -> Q^p_{m n a}):
@substitute!(term04)(\Gamma^p_{m n} -> Q^p_{m n}):

# --- rebuild deriv02

deriv02:=@(term00) + @(term01) + (1/2) @(term02) + (1/6) @(term03) + (1/24) @(term04):

# --- impose RNC conditions

@substitute!(deriv02)(Q^p_{m n} -> 0):
@substitute!(deriv02)(B^a B^b Q^p_{a b} -> 0);

# --- Taylor series for deriv03 -----
@distributed(deriv03):

```

```

term00:=@(deriv03):

tmp:=A^{a} \partial_{a}{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term01)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term02:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term02)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term03:=@(tmp):

@substitute!(term00)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term02)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term02)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term02)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term02)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term03)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term03)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term03)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term03)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv03

```

```

deriv03:=@(term00) + @(term01) + (1/2) @(term02) + (1/6) @(term03):

# --- impose RNC conditions

@substitute!(deriv03)(Q^{p}_{m n} -> 0):
@substitute!(deriv03)(B^a B^b B^c Q^{p}_{a b c} -> 0);

# --- Taylor series for deriv04 -----

@distribute!(deriv04):
term00:=@(deriv04):

tmp:=A^{a} \partial_{a}{@(term00)}:
@distribute!(tmp): @prodrule!(tmp): @distribute!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@(tmp):

tmp:=A^{a} \partial_{a}{@(term01)}:
@distribute!(tmp): @prodrule!(tmp): @distribute!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0, \partial_{a}{A^{b}} -> 0):
term02:=@(tmp):

@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term02)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term02)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term02)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv04

deriv04:=@(term00) + @(term01) + (1/2) @(term02):

# --- impose RNC conditions

@substitute!(deriv04)(Q^{p}_{m n} -> 0):
@substitute!(deriv04)(B^a B^b B^c Q^{p}_{a b c} -> 0):

```

```

@substitute!(deriv04)(B^a B^b B^c B^d Q^{p}_{a b c d} -> 0);

# --- Taylor series for deriv05 -----

@distributed!(deriv05):
term00:=@deriv05:

tmp:=A^{a} \partial_{a}{@(term00)}:
@distributed!(tmp): @prodrule!(tmp): @distributed!(tmp):
@substitute!(tmp)(\partial_{a}{B^{b}} -> 0):
term01:=@tmp:

@substitute!(term00)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term00)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term00)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term00)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

@substitute!(term01)(\partial_{a b c}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b c}):
@substitute!(term01)(\partial_{a b}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a b}):
@substitute!(term01)(\partial_{a}{\Gamma^{p}_{m n}} -> Q^{p}_{m n a}):
@substitute!(term01)(\Gamma^{p}_{m n} -> Q^{p}_{m n}):

# --- rebuild deriv05

deriv05:=@(term00) + @(term01):

# --- impose RNC conditions

@substitute!(deriv05)(Q^{p}_{m n} -> 0):
@substitute!(deriv05)(B^a B^b B^c Q^{p}_{a b c} -> 0):
@substitute!(deriv05)(B^a B^b B^c B^d Q^{p}_{a b c d} -> 0):
@substitute!(deriv05)(B^a B^b B^c B^d B^e Q^{p}_{a b c d e} -> 0);

```

$$deriv02 := -A^d Q^a_{bcd} B^b B^c - \frac{1}{2} A^e A^d Q^a_{bcd} B^b B^c - \frac{1}{6} A^f A^e A^d Q^a_{bcfed} B^b B^c - \frac{1}{24} A^g A^f A^e A^d Q^a_{bcgfed} B^b B^c$$

$$\begin{aligned}
deriv03 := & -A^f B^b B^c B^d Q^a_{cdfb} - \frac{1}{2} A^g A^f B^b B^c B^d Q^a_{cdgfb} + A^g A^f B^b B^c B^d Q^a_{bef} Q^e_{cdg} + A^g A^f B^b B^c B^d Q^a_{beg} Q^e_{cdf} \\
& + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{begf} Q^e_{cdh} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{behf} Q^e_{cdg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{bef} Q^e_{cdhg} \\
& + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{behg} Q^e_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{beg} Q^e_{cdhf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a_{beh} Q^e_{cdgf}
\end{aligned}$$

$$\begin{aligned}
deriv04 := & A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{def} + 4 A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efd} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{def} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{deif} \\
& + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{dehf} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{efd} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efid} \\
& + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{efhd} + A^i A^h B^b B^c B^d B^e Q^a{}_{bfh} Q^f{}_{deic} + A^i A^h B^b B^c B^d B^e Q^a{}_{bfi} Q^f{}_{dehc}
\end{aligned}$$

$$deriv05 := A^j B^b B^c B^d B^e B^f Q^a{}_{bcg} Q^g{}_{efjd} + 3 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{efdg} + 6 A^j B^b B^c B^d B^e B^f Q^a{}_{cgb} Q^g{}_{efjd} + 6 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{fgde}$$

--- print the results -----

```
@print["x^a="~@(deriv00)];
@print["\frac{dx^a}{ds}="~@(deriv01)];
@print["\frac{d^2x^a}{ds^2}="~@(deriv02)~"+\BigO{\eps^6}"];
@print["\frac{d^3x^a}{ds^3}="~@(deriv03)~"+\BigO{\eps^6}"];
@print["\frac{d^4x^a}{ds^4}="~@(deriv04)~"+\BigO{\eps^6}"];
@print["\frac{d^5x^a}{ds^5}="~@(deriv05)~"+\BigO{\eps^6}"];
@print["\frac{d^6x^a}{ds^6}="~@(deriv06)~"+\BigO{\eps^6}"];
```

$$x^a = A^a$$

$$\frac{dx^a}{ds} = B^a$$

$$\frac{d^2x^a}{ds^2} = \left(-A^d Q^a{}_{bcd} B^b B^c - \frac{1}{2} A^e A^d Q^a{}_{bcd} B^b B^c - \frac{1}{6} A^f A^e A^d Q^a{}_{bcfed} B^b B^c - \frac{1}{24} A^g A^f A^e A^d Q^a{}_{bcgfed} B^b B^c \right) + \mathcal{O}(\epsilon^6)$$

$$\begin{aligned} \frac{d^3x^a}{ds^3} = & \left(-A^f B^b B^c B^d Q^a{}_{cdfb} - \frac{1}{2} A^g A^f B^b B^c B^d Q^a{}_{cdgfb} + A^g A^f B^b B^c B^d Q^a{}_{bef} Q^e{}_{cdg} + A^g A^f B^b B^c B^d Q^a{}_{beg} Q^e{}_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{begf} Q^e{}_{cdh} \right. \\ & + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{behf} Q^e{}_{cdg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{bef} Q^e{}_{cdhg} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{behg} Q^e{}_{cdf} + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{beg} Q^e{}_{cdhf} \\ & \left. + \frac{1}{3} A^h A^g A^f B^b B^c B^d Q^a{}_{beh} Q^e{}_{cdgf} \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

$$\begin{aligned} \frac{d^4x^a}{ds^4} = & \left(A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{def} + 4 A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efd} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{def} + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{deif} \right. \\ & + \frac{1}{2} A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{dehf} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bcih} Q^a{}_{efd} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bch} Q^a{}_{efid} + 2 A^i A^h B^b B^c B^d B^e Q^f{}_{bci} Q^a{}_{efhd} \\ & \left. + A^i A^h B^b B^c B^d B^e Q^a{}_{bfh} Q^f{}_{deic} + A^i A^h B^b B^c B^d B^e Q^a{}_{bfi} Q^f{}_{dehc} \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

$$\frac{d^5x^a}{ds^5} = (A^j B^b B^c B^d B^e B^f Q^a{}_{bcg} Q^g{}_{efjd} + 3 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{efdg} + 6 A^j B^b B^c B^d B^e B^f Q^a{}_{cgb} Q^g{}_{efjd} + 6 A^j B^b B^c B^d B^e B^f Q^g{}_{bcj} Q^a{}_{fgde}) + \mathcal{O}(\epsilon^6)$$

$$\frac{d^6x^a}{ds^6} = 0 + \mathcal{O}(\epsilon^6)$$

```

# --- imported from connection.lib -----
# gamma(x) = Gamma00 + x^d Gamma01 + x^d x^e Gamma02 + x^d x^e x^f Gamma03 + ...
# Gamma01 = 1! gamma_{,d}      at x=0, symmetrised over {d}
# Gamma02 = 2! gamma_{,de}     at x=0, symmetrised over {d,e} = ( de + ed )/2!
# Gamma03 = 3! gamma_{,def}    at x=0, symmetrised over {d,e,f} = ( def + edf + dfe + ...)/3!

Gamma01:="import connection.lib Gamma01":
@run(Gamma01){"/Users/leo/local/sh/cdbfile"}:

Gamma02:="import connection.lib Gamma02":
@run(Gamma02){"/Users/leo/local/sh/cdbfile"}:

Gamma03:="import connection.lib Gamma03":
@run(Gamma03){"/Users/leo/local/sh/cdbfile"}:

Gamma04:="import connection.lib Gamma04":
@run(Gamma04){"/Users/leo/local/sh/cdbfile"}:

# --- Gamma0* as imported from connection.lib was *not* symmetrised, do so now -----
@sym!(Gamma02){ _{d}, _{e} }: @canonicalise!(%):
@sym!(Gamma03){ _{d}, _{e}, _{f} }: @canonicalise!(%):
@sym!(Gamma04){ _{d}, _{e}, _{f}, _{g} }: @canonicalise!(%):

# --- cancel the 1/n! that comes from @sym(...) -----
tmp:=@(Gamma02): Gamma02:= 2 @(tmp):
tmp:=@(Gamma03): Gamma03:= 6 @(tmp):
tmp:=@(Gamma04): Gamma04:= 24 @(tmp):

# --- substitute into the deriv0* -----
@substitute!(deriv02)(Q^{a}_{ }_{b c d} -> @(Gamma01)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e} -> @(Gamma02)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e f} -> @(Gamma03)):
@substitute!(deriv02)(Q^{a}_{ }_{b c d e f g} -> @(Gamma04)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv03)(Q^{a}_{ }_{b c d} -> @(Gamma01)):
@substitute!(deriv03)(Q^{a}_{ }_{b c d e} -> @(Gamma02)):

```

```

@substitute!(deriv03)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv04)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv04)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv04)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv05)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv05)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv05)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

@substitute!(deriv06)(Q^{a}_{b c d} -> @(Gamma01)):
@substitute!(deriv06)(Q^{a}_{b c d e} -> @(Gamma02)):
@substitute!(deriv06)(Q^{a}_{b c d e f} -> @(Gamma03)):
@distribute!(%): @prodsort!(%): @rename_dummies!(%): @canonicalise!(%):

# =====
#   force all indices on R to be downstairs
# =====

@substitute!(deriv02)(R^{a}_{b c d} -> R_{a b c d},
    R^{a}_{b}_{c d} -> R_{a b c d},
    R^{a}_{b c}_{d} -> R_{a b c d},
    R^{a b}_{c}_{d} -> R_{a b c d},
    R^{a b c}_{d} -> R_{a b c d},
    R^{a}_{b}_{c d} -> R_{a b c d},
    R^{a}_{b}_{c}_{d} -> R_{a b c d},
    \nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
    \nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
    \nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv03)(R^{a}_{b c d} -> R_{a b c d},
    R^{a}_{b}_{c d} -> R_{a b c d},
    R^{a}_{b c}_{d} -> R_{a b c d},
    R^{a b}_{c}_{d} -> R_{a b c d},
    R^{a b c}_{d} -> R_{a b c d},
    R^{a}_{b}_{c d} -> R_{a b c d},

```

```

R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv04)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv05)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

@substitute!(deriv06)(R^{a}_{b c d} -> R_{a b c d},
R_{a}^{b}_{c d} -> R_{a b c d},
R^{a}_{b c}^{d} -> R_{a b c d},
R_{a b}^{c}_{d} -> R_{a b c d},
R_{a b c}^{d} -> R_{a b c d},
R^{a}_{b}^{c d} -> R_{a b c d},
R^{a}_{b}^{c}_{d} -> R_{a b c d},
\nabla^{a}_{R_{b c d e}} -> \nabla_{a}_{R_{b c d e}},
\nabla^{a}_{f}_{R_{b c d e}} -> \nabla_{a f}_{R_{b c d e}},
\nabla^{a}_{f g}_{R_{b c d e}} -> \nabla_{a f g}_{R_{b c d e}}):

```

```

# --- compute the generalised connections -----
{B^{a},x^{a},R_{a b c d},R^{a}_{b c d},\nabla_{a}\{R_{b c d e}\}}::SortOrder.

@prodsort!(deriv02):
@prodsort!(deriv03):
@prodsort!(deriv04):
@prodsort!(deriv05):
@prodsort!(deriv06):

@rename_dummies!(deriv02):
@rename_dummies!(deriv03):
@rename_dummies!(deriv04):
@rename_dummies!(deriv05):
@rename_dummies!(deriv06):

@canonicalise!(deriv02):
@canonicalise!(deriv03):
@canonicalise!(deriv04):
@canonicalise!(deriv05):
@canonicalise!(deriv06):

# --- raise index {a} and set A^a -> x^a

genGamma02:= - @(deriv02)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma02) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!%(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @factor_out!!%(B^{a}):
genGamma02:= @(tmp); "genGamma02.del"

genGamma03:= - @(deriv03)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma03) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!%(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @factor_out!!%(B^{a}):
genGamma03:= @(tmp); "genGamma03.del"

genGamma04:= - @(deriv04) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma04) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):

```

```
@substitute!(%)(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @factor_out!!(%)(B^{a}):
genGamma04:= @(tmp); "genGamma04.del"
```

```
genGamma05:= - @(deriv05)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma05) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!(%)(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @factor_out!!(%)(B^{a}):
genGamma05:= @(tmp); "genGamma05.del"
```

```
genGamma06:= - @(deriv06)) g^{a u1}:
@distribute!(%): @eliminate_metric!(%):
tmp:=@(genGamma06) \delta^{a}_{u1}: @distribute!(%): @eliminate_kr!(%):
@substitute!(%)(A^{a}->x^{a}): @prodsort!(%): @rename_dummies!(%): @factor_out!!(%)(B^{a}):
genGamma06:= @(tmp); "genGamma06.del"
```

--- the generalised connections -----

```
@print["\genGammaA=~@(\genGamma02)];
@print["\genGammaB=~@(\genGamma03)];
@print["\genGammaC=~@(\genGamma04)];
@print["\genGammaD=~@(\genGamma05)];
@print["\genGammaE=~@(\genGamma06)];
```

$$\begin{aligned}
B^b B^c \Gamma_{bc}^a(x) = & B^b B^c \left(\frac{2}{3} x^d R^a{}_{bdc} + \frac{1}{6} x^d x^e \nabla_b R^a{}_{dec} + \frac{1}{3} x^d x^e \nabla_d R^a{}_{bec} + \frac{1}{12} x^d x^e \nabla^a R_{dbec} + \frac{8}{45} x^d x^e x^f R^a{}_{deg} R_{fbcg} - \frac{4}{45} x^d x^e x^f R^a{}_{bdg} R_{ecfg} - \frac{2}{45} x^d x^e x^f R^a{}_{dbg} R_{ecfg} \right. \\
& + \frac{1}{10} x^d x^e x^f \nabla_{db} R^a{}_{efc} + \frac{1}{10} x^d x^e x^f \nabla_{de} R^a{}_{bfc} - \frac{2}{45} x^d x^e x^f R^a{}_{gdb} R_{ecfg} + \frac{1}{20} x^d x^e x^f \nabla_d R_{ebfc} + \frac{4}{45} x^d x^e x^f x^g R_{dbch} \nabla_e R^a{}_{fgh} \\
& + \frac{1}{30} x^d x^e x^f x^g R^a{}_{deh} \nabla_b R_{fcgh} + \frac{4}{45} x^d x^e x^f x^g R^a{}_{deh} \nabla_f R_{gbch} - \frac{2}{45} x^d x^e x^f x^g R^a{}_{bdh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R^a{}_{dbh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_c R^a{}_{fgh} \\
& - \frac{2}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{cgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{gch} + \frac{1}{30} x^d x^e x^f x^g \nabla_{deb} R^a{}_{fgc} + \frac{1}{45} x^d x^e x^f x^g \nabla_{def} R^a{}_{bgc} - \frac{1}{36} x^d x^e x^f x^g R^a{}_{deh} \nabla_h R_{fbgc} \\
& \left. - \frac{1}{45} x^d x^e x^f x^g R^a{}_{hdb} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla^a R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{hgc} + \frac{1}{60} x^d x^e x^f x^g \nabla_{de} R_{fbgc} \right)
\end{aligned}$$

$$\begin{aligned}
B^b B^c B^d \Gamma_{bcd}^a(x) = & B^b B^c B^d \left(\frac{1}{2} x^e \nabla_b R^a{}_{ced} + \frac{2}{15} x^e x^f R^a{}_{ebg} R_{fcgd} + \frac{8}{15} x^e x^f R^a{}_{beg} R_{fcgd} - \frac{2}{15} x^e x^f R^a{}_{bcg} R_{edfg} + \frac{3}{10} x^e x^f \nabla_{eb} R^a{}_{cfd} + \frac{1}{10} x^e x^f \nabla_{bc} R^a{}_{efd} \right. \\
& + \frac{2}{5} x^e x^f R^a{}_{geb} R_{fcgd} + \frac{1}{20} x^e x^f \nabla_b R_{ecfd} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_d R^a{}_{fgh} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_h R^a{}_{fgd} + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{dgh} + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{hgd} \\
& + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla^a R_{fdgh} + \frac{1}{9} x^e x^f x^g R^a{}_{beh} \nabla_c R_{fdgh} + \frac{2}{9} x^e x^f x^g R^a{}_{beh} \nabla_f R_{gcdh} - \frac{1}{18} x^e x^f x^g R^a{}_{beh} \nabla_h R_{fcgd} + \frac{1}{9} x^e x^f x^g R^a{}_{heb} \nabla_c R_{fdgh} \\
& \left. + \frac{2}{9} x^e x^f x^g R^a{}_{heb} \nabla_f R_{gcdh} - \frac{1}{18} x^e x^f x^g R^a{}_{heb} \nabla_h R_{fcgd} \right)
\end{aligned}$$

$$\begin{aligned}
B^b B^c B^d B^e \Gamma_{bcde}^a(x) = & B^b B^c B^d B^e \left(\frac{4}{9} x^f R^a{}_{bcg} R_{fdge} + \frac{1}{9} x^f x^g R^a{}_{bch} \nabla_d R_{fehg} + \frac{2}{9} x^f x^g R^a{}_{bch} \nabla_f R_{gdeh} - \frac{1}{18} x^f x^g R^a{}_{bch} \nabla_h R_{fdge} + \frac{1}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{geh} \right. \\
& + \frac{5}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{hge} + \frac{2}{9} x^f x^g R_{fbch} \nabla_g R^a{}_{deh} + \frac{4}{9} x^f x^g R_{fbch} \nabla_h R^a{}_{dge} + \frac{1}{9} x^f x^g R_{fbch} \nabla^a R_{gdeh} + \frac{2}{3} x^f x^g R_{fbch} \nabla_d R^a{}_{egh} \\
& \left. + \frac{1}{3} x^f x^g R^a{}_{bfh} \nabla_c R_{gdeh} + \frac{1}{3} x^f x^g R^a{}_{hfb} \nabla_c R_{gdeh} \right)
\end{aligned}$$

$$B^b B^c B^d B^e B^f \Gamma_{bcdef}^a(x) = B^b B^c B^d B^e B^f \left(\frac{2}{3} x^g R^a{}_{bch} \nabla_d R_{gef h} + x^g R_{gbch} \nabla_d R^a{}_{ef h} \right)$$

$$B^b B^c B^d B^e B^f B^g \Gamma_{bcdefg}^a(x) = 0$$

--- terms in the series solution of the geodesic ivp -----

```
deriv02:= - @(genGamma02): @substitute!(deriv02)(x^{a}->A^{a}):
deriv03:= - @(genGamma03): @substitute!(deriv03)(x^{a}->A^{a}):
deriv04:= - @(genGamma04): @substitute!(deriv04)(x^{a}->A^{a}):
deriv05:= - @(genGamma05): @substitute!(deriv05)(x^{a}->A^{a}):
deriv06:= - @(genGamma06): @substitute!(deriv06)(x^{a}->A^{a}):
```

```
@print["x^a="~@(deriv00)];
@print["\frac{dx^a}{ds}="~@(deriv01)];
@print["\frac{d^2x^a}{ds^2}="~@(deriv02)~"+\BigO{\eps^6}"];
@print["\frac{d^3x^a}{ds^3}="~@(deriv03)~"+\BigO{\eps^6}"];
@print["\frac{d^4x^a}{ds^4}="~@(deriv04)~"+\BigO{\eps^6}"];
@print["\frac{d^5x^a}{ds^5}="~@(deriv05)~"+\BigO{\eps^6}"];
@print["\frac{d^6x^a}{ds^6}="~@(deriv06)~"+\BigO{\eps^6}"];
```

$$x^a = A^a$$

$$\frac{dx^a}{ds} = B^a$$

$$\begin{aligned} \frac{d^2x^a}{ds^2} = & (-1) B^b B^c \left(\frac{2}{3} A^d R^a{}_{bdc} + \frac{1}{6} A^d A^e \nabla_b R^a{}_{dec} + \frac{1}{3} A^d A^e \nabla_d R^a{}_{bec} + \frac{1}{12} A^d A^e \nabla^a R_{dbec} + \frac{8}{45} A^d A^e A^f R^a{}_{deg} R_{fbcg} - \frac{4}{45} A^d A^e A^f R^a{}_{bdg} R_{ecfg} - \frac{2}{45} A^d A^e A^f R^a{}_{dbg} R_{ecfg} \right. \\ & + \frac{1}{10} A^d A^e A^f \nabla_{db} R^a{}_{efc} + \frac{1}{10} A^d A^e A^f \nabla_{de} R^a{}_{bfc} - \frac{2}{45} A^d A^e A^f R^a{}_{gdb} R_{ecfg} + \frac{1}{20} A^d A^e A^f \nabla^a{}_d R_{ebfc} + \frac{4}{45} A^d A^e A^f A^g R_{dbch} \nabla_e R^a{}_{fgh} \\ & + \frac{1}{30} A^d A^e A^f A^g R^a{}_{deh} \nabla_b R_{fcgh} + \frac{4}{45} A^d A^e A^f A^g R^a{}_{deh} \nabla_f R_{gbch} - \frac{2}{45} A^d A^e A^f A^g R^a{}_{bdh} \nabla_e R_{fcgh} - \frac{1}{45} A^d A^e A^f A^g R^a{}_{dbh} \nabla_e R_{fcgh} - \frac{1}{45} A^d A^e A^f A^g R_{dbeh} \nabla_c R^a{}_{fgh} \\ & - \frac{2}{45} A^d A^e A^f A^g R_{dbeh} \nabla_f R^a{}_{cgh} - \frac{1}{45} A^d A^e A^f A^g R_{dbeh} \nabla_f R^a{}_{gch} + \frac{1}{30} A^d A^e A^f A^g \nabla_{deb} R^a{}_{fgc} + \frac{1}{45} A^d A^e A^f A^g \nabla_{def} R^a{}_{bgc} - \frac{1}{36} A^d A^e A^f A^g R^a{}_{deh} \nabla_h R_{fbgc} \\ & \left. - \frac{1}{45} A^d A^e A^f A^g R^a{}_{hdb} \nabla_e R_{fcgh} - \frac{1}{45} A^d A^e A^f A^g R_{dbeh} \nabla^a R_{fcgh} - \frac{1}{45} A^d A^e A^f A^g R_{dbeh} \nabla_f R^a{}_{hgc} + \frac{1}{60} A^d A^e A^f A^g \nabla^a{}_{de} R_{fbgc} \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

$$\begin{aligned}
\frac{d^3 x^a}{ds^3} = & (-1) B^b B^c B^d \left(\frac{1}{2} A^e \nabla_b R^a_{ced} + \frac{2}{15} A^e A^f R^a_{ebg} R_{fcdg} + \frac{8}{15} A^e A^f R^a_{beg} R_{fcdg} - \frac{2}{15} A^e A^f R^a_{bcg} R_{edfg} + \frac{3}{10} A^e A^f \nabla_{eb} R^a_{cfd} + \frac{1}{10} A^e A^f \nabla_{bc} R^a_{efd} \right. \\
& + \frac{2}{5} A^e A^f R^a_{geb} R_{fcdg} + \frac{1}{20} A^e A^f \nabla^a_b R_{ecd} + \frac{1}{9} A^e A^f A^g R_{ebch} \nabla_d R^a_{fgh} + \frac{1}{9} A^e A^f A^g R_{ebch} \nabla_h R^a_{fgd} + \frac{2}{9} A^e A^f A^g R_{ebch} \nabla_f R^a_{dgh} + \frac{2}{9} A^e A^f A^g R_{ebch} \nabla_f R^a_{hgd} \\
& + \frac{1}{9} A^e A^f A^g R_{ebch} \nabla^a R_{fdgh} + \frac{1}{9} A^e A^f A^g R^a_{beh} \nabla_c R_{fdgh} + \frac{2}{9} A^e A^f A^g R^a_{beh} \nabla_f R_{gcdh} - \frac{1}{18} A^e A^f A^g R^a_{beh} \nabla_h R_{fcgd} + \frac{1}{9} A^e A^f A^g R^a_{heb} \nabla_c R_{fdgh} \\
& \left. + \frac{2}{9} A^e A^f A^g R^a_{heb} \nabla_f R_{gcdh} - \frac{1}{18} A^e A^f A^g R^a_{heb} \nabla_h R_{fcgd} \right) + \mathcal{O}(\epsilon^6)
\end{aligned}$$

$$\begin{aligned}
\frac{d^4 x^a}{ds^4} = & (-1) B^b B^c B^d B^e \left(\frac{4}{9} A^f R^a_{bcg} R_{fdeg} + \frac{1}{9} A^f A^g R^a_{bch} \nabla_d R_{fegh} + \frac{2}{9} A^f A^g R^a_{bch} \nabla_f R_{gdeh} - \frac{1}{18} A^f A^g R^a_{bch} \nabla_h R_{fdge} + \frac{1}{9} A^f A^g R_{fbch} \nabla_d R^a_{geh} + \frac{5}{9} A^f A^g R_{fbch} \nabla_d R^a_{hge} \right. \\
& \left. + \frac{2}{9} A^f A^g R_{fbch} \nabla_g R^a_{deh} + \frac{4}{9} A^f A^g R_{fbch} \nabla_h R^a_{dge} + \frac{1}{9} A^f A^g R_{fbch} \nabla^a R_{gdeh} + \frac{2}{3} A^f A^g R_{fbch} \nabla_d R^a_{egh} + \frac{1}{3} A^f A^g R^a_{bfh} \nabla_c R_{gdeh} + \frac{1}{3} A^f A^g R^a_{hfb} \nabla_c R_{gdeh} \right) + \mathcal{O}(\epsilon^6)
\end{aligned}$$

$$\frac{d^5 x^a}{ds^5} = (-1) B^b B^c B^d B^e B^f \left(\frac{2}{3} A^g R^a_{bch} \nabla_d R_{gef h} + A^g R_{gbch} \nabla_d R^a_{ef h} \right) + \mathcal{O}(\epsilon^6)$$

$$\frac{d^6 x^a}{ds^6} = 0 + \mathcal{O}(\epsilon^6)$$

--- the series solution to the geodesic initial value problem -----

```
sol:=@(deriv00) + s @(deriv01) + (1/2) s**2 @(deriv02) + (1/6) s**3 @(deriv03)
      + (1/24) s**4 @(deriv04) + (1/120) s**5 @(deriv05) + (1/720) s**6 @(deriv06):
```

```
@substitute!(sol)(A^{a} -> x^a, B^{a}-> \thetdotx^a); "sol.trn"
```

```
@print["\Btag{01}x^{a}(s)="\~@(sol)~"+\BigO{\eps^6}\Etag{01}"];
```

The Taylor series solution to the geodesic equation

$$0 = \frac{d^2 x^a}{ds^2} + \Gamma_{bc}^a(x) \frac{dx^b}{ds} \frac{dx^c}{ds}$$

subject to the initial conditons $x^a(s) = x^a$ and $dx^a/ds = \dot{x}^a$ at $s = 0$, is

$$\begin{aligned} x^a(s) = & \left(x^a + s\dot{x}^a - \frac{1}{2} s^2 \dot{x}^b \dot{x}^c \left(\frac{2}{3} x^d R^a{}_{bdc} + \frac{1}{6} x^d x^e \nabla_b R^a{}_{dec} + \frac{1}{3} x^d x^e \nabla_d R^a{}_{bec} + \frac{1}{12} x^d x^e \nabla^a R_{dbec} + \frac{8}{45} x^d x^e x^f R^a{}_{deg} R_{fbcg} - \frac{4}{45} x^d x^e x^f R^a{}_{bdg} R_{ecfg} \right. \right. \\ & - \frac{2}{45} x^d x^e x^f R^a{}_{dbg} R_{ecfg} + \frac{1}{10} x^d x^e x^f \nabla_{db} R^a{}_{efc} + \frac{1}{10} x^d x^e x^f \nabla_{de} R^a{}_{bfc} - \frac{2}{45} x^d x^e x^f R^a{}_{gdb} R_{ecfg} + \frac{1}{20} x^d x^e x^f \nabla^a{}_d R_{ebfc} + \frac{4}{45} x^d x^e x^f x^g R_{dbch} \nabla_e R^a{}_{fgh} \\ & + \frac{1}{30} x^d x^e x^f x^g R^a{}_{deh} \nabla_b R_{fcgh} + \frac{4}{45} x^d x^e x^f x^g R^a{}_{deh} \nabla_f R_{gbch} - \frac{2}{45} x^d x^e x^f x^g R^a{}_{bdh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R^a{}_{dbh} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_c R^a{}_{fgh} \\ & - \frac{2}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{cgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{gch} + \frac{1}{30} x^d x^e x^f x^g \nabla_{deb} R^a{}_{fgc} + \frac{1}{45} x^d x^e x^f x^g \nabla_{def} R^a{}_{bgc} - \frac{1}{36} x^d x^e x^f x^g R^a{}_{deh} \nabla_h R_{fbgc} \\ & - \frac{1}{45} x^d x^e x^f x^g R^a{}_{hdb} \nabla_e R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla^a R_{fcgh} - \frac{1}{45} x^d x^e x^f x^g R_{dbeh} \nabla_f R^a{}_{hgc} + \frac{1}{60} x^d x^e x^f x^g \nabla^a{}_{de} R_{fbgc} \Big) - \frac{1}{6} s^3 \dot{x}^b \dot{x}^c \dot{x}^d \left(\frac{1}{2} x^e \nabla_b R^a{}_{ced} \right. \\ & + \frac{2}{15} x^e x^f R^a{}_{ebg} R_{fcdg} + \frac{8}{15} x^e x^f R^a{}_{beg} R_{fcdg} - \frac{2}{15} x^e x^f R^a{}_{bcg} R_{edfg} + \frac{3}{10} x^e x^f \nabla_{eb} R^a{}_{cfd} + \frac{1}{10} x^e x^f \nabla_{bc} R^a{}_{efd} + \frac{2}{5} x^e x^f R^a{}_{geb} R_{fcdg} + \frac{1}{20} x^e x^f \nabla^a{}_b R_{ecfd} \\ & + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_d R^a{}_{fgh} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla_h R^a{}_{fgd} + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{dgh} + \frac{2}{9} x^e x^f x^g R_{ebch} \nabla_f R^a{}_{hgd} + \frac{1}{9} x^e x^f x^g R_{ebch} \nabla^a R_{fdgh} + \frac{1}{9} x^e x^f x^g R^a{}_{beh} \nabla_c R_{fdgh} \\ & + \frac{2}{9} x^e x^f x^g R^a{}_{beh} \nabla_f R_{gcdh} - \frac{1}{18} x^e x^f x^g R^a{}_{beh} \nabla_h R_{fcgd} + \frac{1}{9} x^e x^f x^g R^a{}_{heb} \nabla_c R_{fdgh} + \frac{2}{9} x^e x^f x^g R^a{}_{heb} \nabla_f R_{gcdh} - \frac{1}{18} x^e x^f x^g R^a{}_{heb} \nabla_h R_{fcgd} \Big) \\ & - \frac{1}{24} s^4 \dot{x}^b \dot{x}^c \dot{x}^d \dot{x}^e \left(\frac{4}{9} x^f R^a{}_{bcg} R_{fdeg} + \frac{1}{9} x^f x^g R^a{}_{bch} \nabla_d R_{fegh} + \frac{2}{9} x^f x^g R^a{}_{bch} \nabla_f R_{gdeh} - \frac{1}{18} x^f x^g R^a{}_{bch} \nabla_h R_{fdge} + \frac{1}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{geh} + \frac{5}{9} x^f x^g R_{fbch} \nabla_d R^a{}_{hge} \right. \\ & + \frac{2}{9} x^f x^g R_{fbch} \nabla_g R^a{}_{deh} + \frac{4}{9} x^f x^g R_{fbch} \nabla_h R^a{}_{dge} + \frac{1}{9} x^f x^g R_{fbch} \nabla^a R_{gdeh} + \frac{2}{3} x^f x^g R_{fbch} \nabla_d R^a{}_{egh} + \frac{1}{3} x^f x^g R^a{}_{bfh} \nabla_c R_{gdeh} + \frac{1}{3} x^f x^g R^a{}_{hfb} \nabla_c R_{gdeh} \\ & \left. \left. - \frac{1}{120} s^5 \dot{x}^b \dot{x}^c \dot{x}^d \dot{x}^e \dot{x}^f \left(\frac{2}{3} x^g R^a{}_{bch} \nabla_d R_{gef h} + x^g R_{gbch} \nabla_d R^a{}_{efh} \right) \right) \right) + \mathcal{O}(\epsilon^6) \end{aligned}$$

```

# =====
#   export genGamma0*
# =====

com:="open geodesic-ivp.lib":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-ivp.lib genGamma02.del":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-ivp.lib genGamma03.del":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-ivp.lib genGamma04.del":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-ivp.lib genGamma05.del":
@run(com){"/Users/leo/local/sh/cdbfile"}:

com:="export geodesic-ivp.lib genGamma06.del":
@run(com){"/Users/leo/local/sh/cdbfile"}:

# =====
#   export the solution of the ivp for use by truncate.cdbp
# =====

com:="export geodesic-ivp.lib sol.trn":
@run(com){"/Users/leo/local/sh/cdbfile"}:

```